特異性理論からの絵と映画

セルゲイドゥージン
会津大学

平成7年10月31日

0　導入

0.1 この講義では1変数あるいは2変数の関数の特異性についてお話します。そしてMathematicaの機能(記号計算、数値計算、グラフィックス)がどのようにうまく使えるかを見ていくことにします。

0.2 このワークシートは、いかなるMathematicaのワークシートのように、説明のためのテキスト、インプット(命令)とアウトプット(計算結果)から成ります。

それぞれ次のような役目があります。(1) 説明のためのテキストは読むことしかできません。(命令として実行できない。) (2) インプットで書かれた命令を実行できる。実行するために、カーソルをそのラインにおいて、シフトリターンをおして下さい。(3) 命令を実行した結果はアウトプットに含まれます。

0.3 Mathematicaで、「アニメグラフィックス」と言う特別なタイプのアウトプットがある。

今、そのアニメグラフィックスの使い方のドリルをしましょう。

カーソルを下の絵の中にどこかへ入れて、マウスの左のボタンでダブルクリクをしてください。

曲線は、動き始めます。

ウインドウの一番下をみると、家庭用ビデオと同じようなボタンが見える。それらをクリックすることで、アニメーションの速度と順/逆方向をコントロールすることができます。

さあ、やってみましょう！

絵の中で、どこか一度マウスを押せば、アニメーションがとまります。

0.4 この絵の数学的な意味についてはこれからのちに「平面曲線とそれらの特異性」というところで説明します。

0.5 ここで、３Ｄアニメーションの例をあげます。

1　１変数の関数の停留点

1.1 最も簡単な例として、２次の多項式を考えてみましょう

```
f[x_] = x^2 + b x + c
```

係数に値を代入すれば、例えば

```
b = 2; c = 3;
```

その時多項式は、

```
f[x]
```

となる。
この関数のグラフを出してみましょう。

```
Plot[f[x],{x,-4,2}]
```

この簡単なグラフにおいて、特異性の最初の例を見ることができます。

1.2 x 軸の全ての点の中で、他の点と最も違うのは、どの点でしょうか？
この質問に答えるために、与えられたある点の近くでの曲線の行動を見てみましょう。
x 軸の -1 よりも左のグラフ上の点について見るならば、x 軸に沿って左から右へ移動ある場合、曲線は下降あるのが分かります。つまり、減少関数になっています。
一方、-1 よりも右側では、曲線は上昇しています増加関数になっています。

1.3 関数の増加/減少が変わるのは x= -1 だけでする。
$x = -1$ の点は、関数 $f[x] = x^2 + 2x - 3$ の停留点と言われる。

別の用語を用いれば、これは、カタストロフが発生する点と言うこともできます。
特異性理論は、関数の停留点と多様体の特異点を研究する数学の分野である。

1.4 微分学の言い方では、１変数の関数の停留点は、その導関数が０になる点である。
導関数を計算して、そして、それを０にする、x の値を見つけましょう。

```
f1[x_] = D[f[x],x]
Solve[f1[x]==0,x]
```

x の見つけられた値は、上で我々が使用した値です！

1.5 演習。係数ｂとｃにいくつかの値を代入してみましょう。

```
b = ; c = ;
```

多項式のグラフをプロットしましょう。

```
f[x_] = x^2 + b x + c
Plot[f[x],{x,-5,5}]
```

特異点を見つけてから、右と左におけるグラフの行動を説明してみましょう。
Mathematica によって結果をチェックするために、インプットの次の行を使用してみましょう。

```
f1[x] = D[f[x],x]
Solve[f1[x]==0,x]
```

1.6 つぎの式によって、より興味深い例があけられます。

```
f[x_] = x^3 - 3x^2 + 2x
```

この関数のグラフを見ましょう。

```
Plot[f[x],{x,-1,3}]
```

２個の停留点を見ることができます：一つは 0.5 の近くの極大で、もう一つは 1.5 近くの極小です。
次のコマンドはこれらの点の数値を５桁の精度で求めます。

```
NSolve[D[f[x],x]==0,x]
```

1.7 n 次の多項式は、$n-1$ 個の停留点まで持つことが可能です。
次はチェビシェフとよばれる９次の多項式を見ましょう。

```
ChebyshevT[9,x]
Plot[{ChebyshevT[9,x],1,-1},{x,-1.2,1.2}]
```

チェビシェフ多項式は次の注目すべき特徴をもっている。停留点を限度いっぱいもっていること。そしてそれらが２本の水平ライン上に集まっていること。

2　アニメーション：関数族

2.1 パラメーターに依存する関数族があるとしましょ。
パラメーターが変わると、関数の停留点が動いたり、一緒になったり、分かれたり、現れたり、消えたりします。

2.2 次の例では, 関数 x^3+ax を変域 $-2<x<2$ で考える。パラメーター a は $-2<a<1$ の範囲を動く。
0.1 きざみにとった 1 とー 2 の間の数 a に値に対応した x^3+ax の関数グラフをプロットします。そして、それらを集めて映画にしましょう。

```
Do[Plot[ x^3+a*x, {x, -2, 2}, PlotRange ->
{{-2,2}, {-8,8}}], {a, 1, -2, -1/10}]
```

この例において、最初の関数 x^3+x は、停留点を持たない； a=0 になると停留点 (いまの場合, 変曲点) が現れる。その後この点は、２個の異なった単純な（ひとつは極大そしてもうひとつは極小）特異性に分かれる。

2.3 次に挙げるのは、３個の停留点が合体する例です。

```
Do[Plot[ x^4+a*x^2, {x, -2, 2}, PlotRange ->
{{-2,2}, {-6,8}}], {a,-4,1,1/4}]
```

2.4 次のアニメーションの中で、グラフは横方向だけに動きます。
その停留点は、グラフと一諸に動く。

```
Do[Plot[Sin[x-t], {x, 0, 4 Pi}, PlotRange ->
{{0, 4 Pi}, {-2, 2}}, Axes -> False], {t, 0, 2 Pi, Pi/12}]
```

2.5 練習。このグラフが反対の方向（右から左へ）に動き始めるように、前のコマンドを変えてみましょう。
ヒント：単にひとつのシンボルだけ、交換すればよい。

3　平面曲線とそれらの特異性

3.1 今までは、関数の停留点を話をしましたが。
このセクションでは、平面曲線とそれらの特異点を見るだろう。
曲線にある点の近傍でこの曲線が滑らかであれば、その点を通常点と言います。
曲線が滑らかではなければ、その点が特異点と言います。
平面曲線の最も典型的な特異性は、自分自身と交わる点と尖点です。

3.2 例。通常点だけからなる曲線。

```
ParametricPlot[{t^2+1,t*(t^2+1)},{t,-3,3},
PlotRange ->{{-3,3},{-3,3}}]
```

3.3 例。ひとつの尖点をもつ曲線。

```
ParametricPlot[{t^2,t^3},{t,-3,3}, PlotRange
-> {{-3,3},{-3,3}}]
```

3.4 例。一つの自分自身と交わる点を持つ曲線。

```
ParametricPlot[{t^2-1,t*(t^2-1)},{t,-3,3},
PlotRange -> {{-3,3},{-3,3}}]
```

3.5 これらの 3 個の例は、パラメーター表示 $x = t^2 - a$、$y = t(t^2 - a)$ によって与えられた、曲線の一つの族に属しています。

例 3.2 は、$a = -1$、例 3.3 $a = 0$、そして例 3.4 $a = 1$ に相当します。

次のコマンドで、これらの 3 個の曲線を含む映画を生成します。

絵をダブルクリックしてアニメーションをスタートさせましょう。

```
Do[ParametricPlot[{t^2-a,t(t^2-a)},{t,-3,3},
PlotRange -> {{-3,3},{-3,3}}],{a,-1,1,1/6}]
```

3.6 これは 4 個の自分自身と交わる点をもつ曲線です。

```
ParametricPlot[{2 Cos[t]+Cos[5 t],2 Sin[t]+Sin[5 t]},
{t,0,2 Pi},AspectRatio -> 1]
```

3.7 そしてこれは前の曲線を静止画のように含むアニメーションです。

```
Do[ ParametricPlot[{2 Cos[t]+a*Cos[5 t],
2 Sin[t]+a*Sin[5 t]}, {t,0,2 Pi},
PlotRange -> {{-3,3},{-3,3}}, AspectRatio -> 1],
{a,-1,1,0.1}]
```

3.8 練習。t の式を適当にいれて、シフトリターンを押します。

```
x= ; y= ;
```

t の初期値と最終値を決めましょう。

```
t0 = ; t1 = ;
```

プロットコマンドを実行しましょう。

```
ParametricPlot[{x,y},{t,t0,t1}]
```

このようにして得られた曲線の特異性は何でしょうか？

4　２変数の関数

4.1 １変数の関数の特異性理論はとても簡単です。典型的な場合, 極大極小点が x 軸に沿って交互に並んでいます。

２変数の場合は, 安定な特異点は３種類もある。極大、極小、鞍点といいます。これらが平面上に散らばっています。

4.2 3D の絵をよりよくみるために SpinShow という関数を animation パッケージからロードして使いましょう。

```
<<"Graphics`Animation`"
```

次のコマンドを実行しましょう。

```
f[x_,y_] = (x^2 + 3 y^2) Exp[1 - x^2 - y^2]
surf = Plot3D[f[x,y],{x,-2,2},{y,-2.5,2.5}, Boxed -> False,
AxesEdge -> None, PlotPoints -> 21, PlotRange -> {0,3.1}]
```

4.3 ３次元空間内の曲面が得られます。

空間内のどの点からでもながめることもできます。

Plot3D コマンドの ViewPoint オプションで入力する３つ数値により、視点を指定できます。次のコマンドを実行してください。それから、2, -0.8, 0.7 の数を適当に変えて、もういちどそのコマンドを実行してください。

```
Plot3D[f[x,y],{x,-2,2},{y,-2.5,2.5}, Boxed -> False,
AxesEdge -> None, PlotPoints -> 21,
ViewPoint -> {2,-0.8,0.7}]
```

4.4 次のコマンドにより、曲面を動かして見ることができます。

すべての静止画の準備ができたら、メニューの Cell—Closed Group を選びましょう。

```
SpinShow[surf]
```

4.5 これから関数 $f[x,y]$ の停留点について説明します。

これらは、曲面の接平面が水平になる点に対応している。

$f[x,y]$ の例において、こういった点が５個もあります：山頂のような極大が２個、くぼちのような極小が１個と峠のような鞍点が２個であります。

4.6 山の地形図は ContourPlot コマンドで作ることができます。

```
ContourPlot[f[x, y],{x,-2,2},{y,-2.5,2.5},ContourShading ->
True, ColorFunction -> Hue, PlotPoints -> 20,Contours -> 13]
```

4.7 水平方向に切って見ると曲面の様子が良く分かる。方向

次のコマンドは 0 から 3 まで 0.5 きざみで水平断面をとり、それらをフレーム列としたアニメーションをつくりだす。

```
Do[Show[{s, Plot3D[a,{x,-2,2},{y,-2,2}, Boxed -> False,
AxesEdge -> None, BoxRatios -> {2,2,1},PlotRange ->
{0,3.1}, DisplayFunction -> Identity]}],{a,0,3,0.5}]
```

4.8 微分学を使って、２つの偏導関数を０と置く連立方程式を解くことで特異点の座標が分かります。

```
eqn1 = D[f[x,y],x]==0
eqn2 = D[f[x,y],y]==0
Solve[{eqn1,eqn2},{x,y}]
```

上の曲面の絵に対しては、５つの解が得られた。

4.9 ある停留点が極大何のか、極小何のかあるいは鞍点何のか知るために、2 階の導関数を計算して、2×2 の行列にしす 。

```
dxx = D[f[x,y],x,x] /. {x->-1,y->0};
dxy = D[f[x,y],x,y] /. {x->-1,y->0};
dyy = D[f[x,y],y,y] /. {x->-1,y->0};
MatrixForm[{{dxx,dxy},{dxy,dyy}}]
```

この行列の行列式が負の数だから（$det < 0$）、$(-1,0)$ の点は鞍点です。
行列式正であれば（$det > 0$）、$dxx > 0$ 場合には極小、$dxx < 0$ 場合には極大がそれぞれ得られる。

4.10 練習。他の特異点についてこの行列計算をしてみましょう。

4.11 練習。次の関数について今まで行なったコマンドを実行してみましょう。

```
f[x_,y_] = Cos[x]+Cos[y]+Cos[x+y]
s = Plot3D[f[x,y],{x,-Pi,Pi},{y,-Pi,Pi}, Boxed -> False,
AxesEdge -> None]
SpinShow[s]
```

5　特異性をもつ曲面

5.1 この節では特異性をもつ曲面の例をいくつか紹介する。
通常点というのはその近傍がなめらか、つまり平面とそんなに変わらないような状況があるところをいう。そうでないときは特異点。
曲面にある一点は、その点の近傍が滑らかであれば通常点とよんでいます。というのは平面の一部のように見えるわけです。そうではない場合は特異点といわれるわけです。
Cone 状の曲面の頂点は孤立特異点のもっとも簡単な例である。

```
cone=ParametricPlot3D[{u Cos[v],u Sin[v],u},
{u,-1,1},{v,0,2 Pi},Axes->False, Boxed->False]
```

5.2 ２つの平面の交線は孤立していない特異性のもっとも簡単な例である。

```
p1 = Plot3D[-x,{x,-1,1},{y,-1,1},
DisplayFunction->Identity]
p2 = Plot3D[y,{x,-1,1},{y,-1,1},
DisplayFunction->Identity]
Show[{p1,p2},AspectRatio->1,
DisplayFunction->$DisplayFunction]
```

つぎの曲面は同じタイプの特異性をもつ。

```
ParametricPlot3D[{u(u^2-1),v,u^2},
{u,-1.3,1.3},{v,-1,1},Axes->False, Boxed->False]
```

5.3 次の Enneper の曲面はより複雑に見えますが、その特異性はとても簡単です。「平面どうしの交わり」と同じタイプなのです。

```
ParametricPlot3D[{Re[E^(I*th)*r - (E^(3*I*th)*r^3)/3],
Re[I*E^(I*th)*r + I/3*E^(3*I*th)*r^3],
Re[E^(2*I*th)*r^2]}, {th, 0, 2 Pi}, {r, 0, 3},
ViewPoint -> {2.8, -1.9, 0.1}, Boxed->False,
PlotPoints -> {60, 20}, Axes -> False, PlotRange -> All,
LightSources ->{{{0.7,0,0.7}, RGBColor[0.9,0,0]},
{{0.6,0.6,0.6}, RGBColor[0,0.9,0]},
{{0,0.7,0.7}, RGBColor[0,0,1]}}]
```

5.4 最後の例はチェビシェフ多項式から作られた、3つの孤立特異点をもつ曲面です。それは3次の Chmutov 曲面と呼ばれています。